
name:                diagrams-pdf
version:             0.3.1
synopsis:            PDF backend for diagrams drawing EDSL
homepage:            http://www.alpheccar.org
license:             BSD3
license-file:        LICENSE
Extra-source-files:  README.md, CHANGES.md
author:              alpheccar
maintainer:          misc@alpheccar.org
Bug-reports:         http://github.com/alpheccar/diagrams-pdf/issues
Stability:           Experimental
copyright:           2013, alpheccar.org         
category:            Graphics
build-type:          Simple
Tested-with:         GHC == 7.6.3
cabal-version:       >=1.10
Description:         This package provides a modular backend for rendering
                     diagrams created with the diagrams EDSL to PDF
                     files.  It uses @HPDF@ making it suitable for use on
                     any platform.
                     .
                     It is a very preliminary version where only the
                     diagrams Logo generation has been tested.
                     .
                     The package provides the following modules:
                     .
                     * "Diagrams.Backend.Pdf.CmdLine" - if you're
                       just getting started with diagrams, begin here.
                     .
                     * "Diagrams.Backend.Pdf" - look at this next.
                       The general API for the HPDF backend.
                     .
                     Additional documentation can be found in the
                     README file distributed with the source tarball
extra-source-files: test/test.hs, test/Makefile, test/logo.jpg

Source-repository head
  type:     git
  location: http://github.com/alpheccar/diagrams-pdf


library
  Exposed-modules:     Diagrams.Backend.Pdf
                       Diagrams.Backend.Pdf.CmdLine     
  other-modules:       Diagrams.Backend.Pdf.Specific
  build-depends:       base >= 4.6 && < 4.8,
                       mtl >= 2.1 && < 2.2 ,
                       monoid-extras >= 0.3 && < 0.4,
                       semigroups >= 0.9.2 && < 0.10,
                       HPDF >= 1.4.5 && < 1.5,
                       vector-space >= 0.8.6 && < 0.9,
                       diagrams-lib >= 0.7 && < 0.8,
                       diagrams-core >= 0.7 && < 0.8,
                       filepath >= 1.3 && < 1.4,
                       split >= 0.2.2 && < 0.3,
                       cmdargs >= 0.10 && < 0.11,
                       colour >= 2.3.3 && < 2.4,
                       lens >= 3.9 && < 3.10
  Hs-source-dirs:      src

  Ghc-options:         -Wall -fno-warn-unused-binds

diagrams-pdf-0.3.1/LICENSE


Copyright 2013 alpheccar.org

All rights reserved.

Redistribution and use in source and binary forms, with or without
modification, are permitted provided that the following conditions are met:

    * Redistributions of source code must retain the above copyright
      notice, this list of conditions and the following disclaimer.

    * Redistributions in binary form must reproduce the above
      copyright notice, this list of conditions and the following
      disclaimer in the documentation and/or other materials provided
      with the distribution.

    * Neither the name of Ryan Yates nor the names of other
      contributors may be used to endorse or promote products derived
      from this software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS
"AS IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT
LIMITED TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR
A PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT
OWNER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL,
SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT
LIMITED TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE,
DATA, OR PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY
THEORY OF LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT
(INCLUDING NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE
OF THIS SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.







diagrams-pdf-0.3.1/README.md


_diagrams-pdf_ is a an [HPDF] backend for [diagrams]. Diagrams is a powerful,
flexible, declarative domain-specific language for creating vector graphics,
using the [Haskell programming language][haskell].

[diagrams]: http://projects.haskell.org/diagrams/
[haskell]: http://www.haskell.org/haskellwiki/Haskell
[HPDF]: http://hackage.haskell.org/packages/archive/HPDF/1.4.5/doc/html/Graphics-PDF-Documentation.html

_diagrams-pdf_ is a work in progress, and some features are not implemented yet. However, it 
is functional enough. 

# Installation

```
cabal update && cabal install diagrams-pdf
```

# Usage

A simple example that uses _diagrams-pdf_ to draw a square.

``` haskell
import Diagrams.Prelude
import Diagrams.Backend.Pdf.CmdLine

b1 = square 20 # lw 0.002

main = defaultMain (pad 1.1 b1)
```

Save this to file named `Square.hs` and compile this program:

```
ghc --make Square.hs
```

This will generate an executable which, when run produces a Pdf file. Run the
executable with the `--help` option to find out more about how to call it.

```
$ ./Square --help
Command-line diagram generation.

Square [OPTIONS]

Common flags:
  -w --width=INT    Desired width of the output image (default 400)
  -h --height=INT   Desired height of the output image (default 400)
  -o --output=FILE  Output file
  -c --compressed   Compressed PDF file
  -? --help         Display help message
  -V --version      Print version information
```

You _must_ pass an output file name with a `.pdf` extension to generate the PDF
file.

```
$ ./Square -o square.pdf
```

# Limitations

Two goals of the HPDF library were : powerful typesetting and no dependence on the OS libraries (for better portability).
But, HPDF was also my first big library mainly done to learn. So I had to limit myself.

As a consequence, HPDF is only using the standard fonts defined in the PDF specification. It is not (yet) possible
to include and use any other fonts.

So, for diagrams it is causing a problem : the face setting has nearly no meaning.
Face is currently selected from the Italic and Oblique settings. Times is the default but when
oblique is chosen, Helvetica is used.

I hope to improve this in future releases.

Here is a list of the fonts in HPDF:

* Helvetica	 
* Helvetica_Bold	 
* Helvetica_Oblique	 
* Helvetica_BoldOblique	 
* Times_Roman	 
* Times_Bold	 
* Times_Italic	 
* Times_BoldItalic	 
* Courier	 
* Courier_Bold	 
* Courier_Oblique	 
* Courier_BoldOblique	 
* Symbol	 
* ZapfDingbats







diagrams-pdf-0.3.1/Setup.hs

import Distribution.Simple
main = defaultMain







diagrams-pdf-0.3.1/src/Diagrams/Backend/Pdf.hs

{-# LANGUAGE DeriveDataTypeable        #-}
{-# LANGUAGE FlexibleContexts          #-}
{-# LANGUAGE FlexibleInstances         #-}
{-# LANGUAGE GADTs                     #-}
{-# LANGUAGE MultiParamTypeClasses     #-}
{-# LANGUAGE NoMonomorphismRestriction #-}
{-# LANGUAGE TypeFamilies              #-}
{-# LANGUAGE TypeSynonymInstances      #-}
{-# LANGUAGE ViewPatterns              #-}
{-# LANGUAGE GeneralizedNewtypeDeriving #-}
{-# LANGUAGE Rank2Types #-}
{-# LANGUAGE UndecidableInstances #-}
{-# LANGUAGE TemplateHaskell #-}
-----------------------------------------------------------------------------
-- |
-- Module      :  Diagrams.Backend.Pdf
-- Copyright   :  (c) 2013 alpheccar.org (see LICENSE)
-- License     :  BSD-style (see LICENSE)
--
-- A Pdf rendering backend for diagrams.
--
-- To build diagrams for Pdf rendering use the @Pdf@
-- type in the diagram type construction
--
-- > d :: Diagram Pdf R2
-- > d = ...
--
-- and render giving the @Pdf@ token
--
-- > renderDia Pdf (PdfOptions (Width 400)) d
--
-- This IO action will write the specified file.
--
-- / Specific HPDF primitives /
-- 
-- For details about the use of the HPDF specific primitives, the file
-- test.hs in this package can be used. You'll have to unpack the archive
-- for this package.
-----------------------------------------------------------------------------
module Diagrams.Backend.Pdf

  ( -- * PDF Backend 
    -- ** Backend token
    Pdf(..)
    -- ** Backend options
  , Options(..)
  , sizeFromSpec
  -- * HPDF Specific primitives 
  -- ** Text
  , LabelStyle(..)
  , TextOrigin(..)
  , LabelSize
  , pdfLabelWithSuggestedSize
  , pdfTextWithSuggestedSize
  , pdfLabelWithSize
  , pdfTextWithSize
  -- ** Image
  , pdfImage
  -- ** URL
  , pdfURL
  -- ** Shading
  , pdfAxialShading
  , pdfRadialShading
  ) where


import  Graphics.PDF hiding(transform,Style,translate,scale)
import qualified Graphics.PDF as P

import           Diagrams.Prelude

import           Diagrams.TwoD.Text

import           Data.Maybe                    (catMaybes)


import qualified Data.Foldable                 as F
import           Data.Monoid.Split
import qualified Control.Monad.State.Strict as S
import Control.Monad.Trans(lift)
import Diagrams.TwoD.Path 
import Control.Monad(when)
import Diagrams.Backend.Pdf.Specific
import           Data.Typeable
import qualified Diagrams.TwoD.Shapes as Sh
import Data.Maybe(isJust)
import Control.Lens hiding(transform,(#),para)

--import Debug.Trace as T

--debug a = T.trace (show a) a 

-- | This data declaration is simply used as a token to distinguish this rendering engine.
data Pdf = Pdf
    deriving (Eq,Ord,Read,Show,Typeable)

data FillingMode = NoFilling | Shading | ColorFilling deriving(Eq,Show)

-- | The drawing state
-- I should give a name to the different fields and use lens
-- The first three parameters are for the font
-- The P.Point is because the PDF specification is using absolute
-- coordinates but relative coordinates are needed for diagrams.
-- So the last point is tracked here.
-- The last bools are to disable fill and stroke.
-- Fill is disabled when transparency is total (color transparency and not diagram transparency)
-- Stroke is disabled when line width is 0 because in the PDF specification
-- line width 0 means the smallest width and something is displayed.
data DrawingState = DrawingState { _fontSlant :: FontSlant
                                 , _fontWeight :: FontWeight
                                 , _fontSize :: Int
                                 , _fillRule :: FillRule 
                                 , _currentPoint :: P.Point 
                                 , _fillingMode :: FillingMode
                                 , _mustStroke :: Bool
                                 , _isloop :: Bool
                                 , _shading :: Maybe PDFShading
                                 , _strokeOpacity :: Double 
                                 , _fillOpacity :: Double
                               }
makeLenses ''DrawingState

-- | The stack of drawing state
data StateStack = StateStack { _current :: DrawingState
                             , _last :: [DrawingState]
                             }
makeLenses ''StateStack

defaultFontSize :: Num a => a
defaultFontSize = 1 

diagramDefaultUnit :: Fractional a => a 
diagramDefaultUnit = 0.01 

defaultWidth :: Fractional a => a 
defaultWidth = 0.01

-- | Initial drawing state
initState :: StateStack
initState = StateStack (DrawingState FontSlantNormal FontWeightNormal 1 Winding (0 :+ 0) NoFilling True True Nothing 1.0 1.0) []


-- | The drawing monad with state
newtype DrawS a = DS (S.StateT StateStack Draw a) deriving(Monad,S.MonadState StateStack)

-- | List a Draw value into the DrawS monad
drawM :: Draw a -> DrawS a
drawM = DS . lift


-- | Get the MonadState wrapped in DS
unDS :: DrawS t -> S.StateT StateStack Draw t
unDS (DS r) = r 

-- | Get the Draw value from an initial state
runDS :: DrawS a -> Draw a
runDS d = S.evalStateT (unDS d) initState

-- | Generate an HPDF font
mkFont :: (FontSlant,FontWeight,Int) -> PDFFont 
mkFont (FontSlantNormal, FontWeightNormal, s) = PDFFont Times_Roman s
mkFont (FontSlantNormal, FontWeightBold, s) = PDFFont Times_Bold s
mkFont (FontSlantItalic, FontWeightNormal, s) = PDFFont Times_Italic s
mkFont (FontSlantItalic, FontWeightBold, s) = PDFFont Times_BoldItalic s
mkFont (FontSlantOblique, FontWeightNormal, s) = PDFFont Helvetica_Oblique s
mkFont (FontSlantOblique, FontWeightBold, s) = PDFFont Helvetica_BoldOblique s

extractFont :: DrawingState -> (FontSlant,FontWeight,Int)
extractFont s = (s ^. fontSlant, s ^. fontWeight, s ^. fontSize)

setFontSize :: Double -> DrawS ()
setFontSize fs = do 
  let s = floor fs
  current . fontSize .= s
  

setFontWeight :: FontWeight -> DrawS ()
setFontWeight w = current . fontWeight .= w 

setFontSlant :: FontSlant -> DrawS ()
setFontSlant sl = current . fontSlant .= sl 
  
setFillRule :: FillRule -> DrawS ()
setFillRule wr = current . fillRule .= wr 

savePoint :: P.Point -> DrawS () 
savePoint p = current . currentPoint .= p

-- | From the alpha value of a fill color, we check if the filling must be disabled
checkFillingNeeded :: Double -> DrawS() 
checkFillingNeeded alpha = do 
    let b | alpha /= 0.0 = id 
          | otherwise = const NoFilling
    current . fillingMode %= b

-- | From the linew width we check if stroke must be disabled
setTrokeState :: Double -> DrawS () 
setTrokeState w = do 
  let st | w == 0 = False 
         | otherwise = True
  current . mustStroke .= st
 
isALoop :: DrawS Bool 
isALoop = use (current . isloop)

getShading :: DrawS (Maybe PDFShading) 
getShading = use (current . shading)

setLoop :: Bool -> DrawS () 
setLoop b = current . isloop .= b 

-- | Initial settings before rendering the diagram
setTranform :: Draw () -> Draw ()
setTranform d = do
     P.fillColor P.white 
     P.strokeColor P.black
     P.setWidth defaultWidth
     d

withShading :: Transformation R2 
            -> FillRule 
            -> FillingMode 
            -> Bool -- ^ Must stroke
            -> Maybe PDFShading 
            -> DrawS () 
            -> DrawS ()
withShading td evenodd Shading strokedRequested (Just shade)  drawCommands = do 
    withContext $ do 
      drawCommands
      drawM $ do
         if evenodd == EvenOdd
         then P.setAsClipPathEO
         else P.setAsClipPath
         P.applyShading (unTrans $ transform td (TransSh shade))
    when (strokedRequested) $ do 
      drawCommands 
      drawM (P.strokePath) 
withShading _ _ Shading strokedRequested Nothing drawCommands = 
    when (strokedRequested) $ do 
      drawCommands 
      drawM (P.strokePath)
withShading _ Winding ColorFilling True _ drawCommands = do 
  drawCommands 
  drawM P.fillAndStrokePath
withShading _ EvenOdd ColorFilling True _ drawCommands = do 
  drawCommands 
  drawM P.fillAndStrokePathEO
withShading _ Winding ColorFilling False _ drawCommands = do 
  drawCommands 
  drawM P.fillPath
withShading _ EvenOdd ColorFilling False _ drawCommands = do 
  drawCommands 
  drawM P.fillPathEO
withShading _ _ NoFilling True _ drawCommands = do 
  drawCommands 
  drawM P.strokePath
withShading _ _ NoFilling False _ _ = return ()

strokeOrFill :: Transformation R2 -> DrawS () -> DrawS () 
strokeOrFill td r = do
  let whenNoLoop m = do
        islooppath <- isALoop
        if islooppath
          then m
          else return NoFilling 
  fm <- whenNoLoop $ use (current . fillingMode)
  sm <- use (current . mustStroke) 
  fr <- use (current . fillRule)
  sh <- getShading
  withShading td fr fm sm sh r 
  setLoop True
  return ()

-- | Perform a rendering operation with a local style.
withStyle'     :: Style R2    -- ^ Style to use
               -> Transformation R2  -- ^ Transformation to be applied to the style
               -> Transformation R2 -- ^ Transformation that was applied to the diagram
               -> Render Pdf R2 -- ^ Rendering operation to run
               -> Render Pdf R2 -- ^ Rendering operation using the style locally
withStyle' s t td (D r) = D $ do
    withContext $ do
       pdfMiscStyle s
       -- Set the clip region into a new PDF context
       -- since it is the only way to restore the old clip region
       -- (by popping the PDF stack of contexts)
       pdfTransf t
       withClip t s $ do
          pdfFrozenStyle s
          diagramOpacity <- getDiagramOpacity s
          use (current . strokeOpacity) >>= drawM . P.setStrokeAlpha . (* diagramOpacity)
          use (current . fillOpacity) >>= drawM . P.setFillAlpha . (* diagramOpacity)
          strokeOrFill (td) r

instance Backend Pdf R2 where
  data Render  Pdf R2 = D (DrawS ())
  type Result  Pdf R2 = Draw ()
  data Options Pdf R2 = PdfOptions {
      pdfsizeSpec :: SizeSpec2D
    } deriving(Show)
     
  -- There is something I don't understand here with the frozen style.
  -- On the tests it is working but I would not have put
  -- the calls in this order ... so it must be checked later   
  withStyle _ s t (D r) = withStyle' s t mempty (D r)

  doRender _ _ (D r) = setTranform (runDS r)

  renderDia Pdf opts d =
    centerAndScale d' . doRender Pdf opts' . mconcat . map renderOne . prims $ d'
      where (opts', d') = adjustDia Pdf opts d
            renderOne :: (Prim Pdf R2, (Split (Transformation R2), Style R2))
                      -> Render Pdf R2
            renderOne (p, (M t,      s))
              = withStyle' s mempty t (render Pdf (transform t p))

            renderOne (p, (t1 :| t2, s))
              -- Here is the difference from the default
              -- implementation: "t2" instead of "t1 <> t2".
              = withStyle' s t1 t2 (render Pdf (transform t2 p))
            centerAndScale diag renderedDiagram  = do
                let bd = boundingBox diag
                    (w,h) = sizeFromSpec (pdfsizeSpec opts)
                    rescaledD (Just (ll,ur)) =
                                let (vx,vy) = unp2 $ centroid [ll,ur]
                                    (xa,ya) = unp2 ll 
                                    (xb,yb) = unp2 ur 
                                    --ps = max (abs (xb - xa)) (abs (yb - ya))
                                    sx = w / (abs (xb - xa))
                                    sy = h / abs (yb - ya)
                                    s = min sx sy
                                    pageCenter = (w / 2.0) P.:+ (h/2.0)
                                in
                                do
                                  P.applyMatrix (P.translate pageCenter) 
                                  P.applyMatrix (P.scale s s)
                                  P.applyMatrix (P.translate $ (-vx) P.:+ (-vy))
                    rescaledD Nothing = return ()
                rescaledD (getCorners bd)
                P.withNewContext $ do
                  renderedDiagram

instance Monoid (Render Pdf R2) where
  mempty  = D (return ())
  (D a) `mappend` (D b) = D (a >> b)

sizeFromSpec :: SizeSpec2D -> (Double,Double)    
sizeFromSpec size = case size of
   Width w'   -> (w',w')
   Height h'  -> (h',h')
   Dims w' h' -> (w',h')
   Absolute   -> (200,200)

-- | Relative lineto
relativeLine :: P.Point -> DrawS ()
relativeLine p = do 
  c <- use (current . currentPoint) 
  let c' = p + c
  drawM (lineto c')
  savePoint c'

-- | Relative curveto
relativeCurveTo :: P.Point -> P.Point -> P.Point -> DrawS () 
relativeCurveTo x y z = do 
  c <- use (current . currentPoint) 
  let x' = x + c 
      y' = y + c 
      z' = z + c 
  drawM (curveto x' y' z')
  savePoint z'

-- | moveto but with saving of the point
moveToAndSave :: P.Point -> DrawS () 
moveToAndSave p = do 
  drawM (moveto p)
  savePoint p

-- | Convenience functions
renderC :: (Renderable a Pdf, V a ~ R2) => a -> DrawS ()
renderC a = case render Pdf a of D r -> r

-- | With a new context do something
-- It is a bit comlex because the withNewContext from the HPDF library
-- must be used to push / pop a new PDF context
withContext :: DrawS a -> DrawS a
withContext d = do 
  s <- S.get
  let d' = S.evalStateT (unDS d) s
  a <- drawM (withNewContext d') 
  return a



pdfFillColor :: (Real b, Floating b) => AlphaColour b -> DrawS ()
pdfFillColor c = do
  let (r,g,b,a) = colorToSRGBA c
  drawM $ do
      P.setFillAlpha a
      P.fillColor (Rgb r g b)
  current . fillingMode .= ColorFilling
  checkFillingNeeded a
  current . fillOpacity .= a

pdfStrokeColor :: (Real b, Floating b) => AlphaColour b -> DrawS ()
pdfStrokeColor c = do
  let (r,g,b,a) = colorToSRGBA c
  drawM $ do
     P.setStrokeAlpha a
     P.strokeColor (Rgb r g b)
  current . strokeOpacity .= a

setShadingData :: Maybe PDFShading -> DrawS () 
setShadingData sh = do 
  current . shading .= sh 
  current . fillingMode %= \x -> if isJust sh then Shading else x

setShading :: PdfShadingData -> DrawS ()
setShading (PdfAxialShadingData pa pb ca cb) = do 
  let (ra,ga,ba,_) = colorToSRGBA ca
      (rb,gb,bb,_) = colorToSRGBA cb 
      colora = Rgb ra ga ba 
      colorb = Rgb rb gb bb
      (xa,ya) = unp2 pa 
      (xb,yb) = unp2 pb 
  setShadingData $ Just (AxialShading xa ya xb yb colora colorb)
setShading (PdfRadialShadingData pa radiusa pb radiusb ca cb) = do 
  let (ra,ga,ba,_) = colorToSRGBA ca
      (rb,gb,bb,_) = colorToSRGBA cb 
      colora = Rgb ra ga ba 
      colorb = Rgb rb gb bb
      (xa,ya) = unp2 pa 
      (xb,yb) = unp2 pb 
  setShadingData $ Just $ RadialShading xa ya radiusa xb yb radiusb colora colorb
{-

Conversions between diagrams and HPDF types
  
-}

pdfLineJoin :: LineJoin -> P.JoinStyle
pdfLineJoin LineJoinMiter = MiterJoin
pdfLineJoin LineJoinRound = RoundJoin
pdfLineJoin LineJoinBevel = BevelJoin

pdfLineCap :: LineCap -> P.CapStyle 
pdfLineCap LineCapButt = ButtCap 
pdfLineCap LineCapRound = RoundCap
pdfLineCap LineCapSquare = SquareCap

pdfDashing :: Dashing -> DashPattern 
pdfDashing (Dashing l a) = DashPattern (map convert l) (convert a)
  where 
    convert x = defaultWidth * x / diagramDefaultUnit

{-

Opacity and clip attributes must be handled separately from the other
attributes

-}

getDiagramOpacity :: Style v -> DrawS Double
getDiagramOpacity s = do 
     let mo = handle s getOpacity
     case mo of 
      Nothing -> return 1.0 
      Just d -> return d

  where handle :: AttributeClass a => Style v -> (a -> b) -> Maybe b
        handle st f = f `fmap` getAttr st

withClip :: Transformation R2 -> Style v -> DrawS () -> DrawS ()
withClip t s m = do 
     let d = handle s m
     case d of 
       Just r -> r 
       Nothing -> m

  where handle :: Style v -> DrawS () -> Maybe (DrawS ())
        handle st dm = (clipPath dm . getClip) `fmap` getAttr st
        addPathToClip p = do 
          renderC p 
          f <- use (current . fillRule) 
          case f of 
               Winding -> drawM (setAsClipPath)
               EvenOdd -> drawM (setAsClipPathEO)
        clipPath dm p = do 
          withContext $ do 
            pdfTransf ( inv t)
            mapM_ addPathToClip p 
            pdfTransf (t)
            dm


pdfMiscStyle :: Style v -> DrawS ()
pdfMiscStyle s = do
  sequence_ . catMaybes $ [ handle fSlant
                          , handle fWeight
                          , handle fSize
                          , handle fColor
                          , handle lColor
                          , handle lFillRule
                          , handle checklWidth
                          , handle theShading
                          ]
  where handle :: AttributeClass a => (a -> DrawS ()) -> Maybe (DrawS ())
        handle f = f `fmap` getAttr s
        fSize    = setFontSize . getFontSize
        --fFace    = const (return ())
        fSlant   = setFontSlant  . getFontSlant
        fWeight  = setFontWeight . getFontWeight
        lColor c = pdfStrokeColor . toAlphaColour . getLineColor $ c
        fColor c = pdfFillColor . toAlphaColour . getFillColor $ c
        lFillRule = setFillRule . getFillRule
        theShading = setShading . getShadingData
        checklWidth w = do 
          let d = getLineWidth w
          setTrokeState d

pdfFrozenStyle :: Style v -> DrawS ()
pdfFrozenStyle s = sequence_ -- foldr (>>) (return ())
              . catMaybes $ [ handle lWidth
                            , handle lJoin
                            , handle lCap
                            , handle lDashing
                            ]
  where handle :: (AttributeClass a) => (a -> DrawS ()) -> Maybe (DrawS ())
        handle f = f `fmap` getAttr s
        lWidth w = do 
          let d = getLineWidth w
          drawM . setWidth $ (defaultWidth * d / diagramDefaultUnit)
          setTrokeState d
        lCap = drawM . setLineCap . pdfLineCap . getLineCap
        lJoin = drawM . setLineJoin . pdfLineJoin . getLineJoin
        lDashing = drawM . setDash . pdfDashing . getDashing
       
unR :: R2 -> Complex Double
unR r = let (x,y) = unr2 r
 in (x :+ y)

unP :: P2 -> Complex Double
unP r = let (x,y) = unp2 r
 in (x :+ y)

--showTrans :: Transformation R2 -> String 
--showTrans t = show a1 ++ " " ++ show b1 ++ " " ++ show c1 ++ "\n" ++
--              show a2 ++ " " ++ show b2 ++ " " ++ show c2 ++ "\n"
--   where (a1,a2) = unr2 $ apply t unitX
--         (b1,b2) = unr2 $ apply t unitY
--         (c1,c2) = unr2 $ transl t

--instance Show (Transformation R2) where 
--  show = showTrans

pdfTransf :: Transformation R2 -> DrawS ()
pdfTransf t = drawM $ applyMatrix (Matrix a1 a2 b1 b2 c1 c2)
  where (a1,a2) = unr2 $ apply t unitX
        (b1,b2) = unr2 $ apply t unitY
        (c1,c2) = unr2 $ transl t

instance Renderable (Segment Closed R2) Pdf where
  render _ (Linear (OffsetClosed (unR -> v))) = D $ relativeLine v
  render _ (Cubic (unR -> po1)
                  (unR -> po2)
                  (OffsetClosed (unR -> po3)))
    = D $ relativeCurveTo po1 po2 po3

instance Renderable (Trail R2) Pdf where
  render _ t = D . flip withLine t $ renderT . lineSegments
    where
      renderT segs =
        do
          mapM_ renderC segs
          when (isLoop t) (drawM closePath)
          setLoop (isLoop t)

instance Renderable (Path R2) Pdf where
  render _ (Path t) = D $ do
    F.mapM_ renderTrail t
    where renderTrail (viewLoc -> (unP -> p, tr)) = do
            moveToAndSave p
            renderC tr

instance Renderable Text Pdf where
  render _ (Text tr al str) = 
      D $ withContext $ do
            StateStack f _ <- S.get 
            let theFont = mkFont . extractFont $ f
                tw = textWidth theFont (toPDFString str) 
                descent = getDescent theFont 
                fontHeight = getHeight theFont 
                (x,y) = case al of
                    BoxAlignedText xt yt -> (xt,yt)
                    BaselineText         -> (0,0)
                x' = - tw * x 
                y' = - (fontHeight - descent) * y
            pdfTransf tr
            withContext . drawM $ do
              P.applyMatrix (P.scale (1.0 / defaultFontSize) (1.0 / defaultFontSize))
              P.applyMatrix (P.translate (x' :+ y'))
              P.drawText $ P.text theFont 0 0 (toPDFString str)
            
{-

Rendering of specific HPDF primitives

-}

instance Renderable PdfTextBox Pdf where
  render _ (PdfTextBox t w h para) = D $ do
    withContext $ do
       pdfTransf t
       drawM (drawStringLabel w h para)

-- | Typeset a text with a given style in a suggested box.
-- The function is returning a diagram for the typeset text
-- and a diagram for the bounding box which may be smaller
-- than the suggested size : smaller width when the algorithm
-- has done some line justification. 
-- The text may also be bigger than the suggested width in case
-- of overflow (similar to the way TeX is doing thing. There are
-- settings in HPDF to control the elegance of the line cuts but
-- those settings are not accessible from this simple API).
-- The text will not be longer than the suggested height. In that
-- case the additional text is not displayed.
genericPdfText :: (Renderable PdfTextBox Pdf,Renderable (Path R2) Pdf) 
               => Bool -- ^ Suggested size 
               -> TextOrigin
               -> Double -- ^ Suggested width
               -> Double -- ^ Suggested height
               -> AnyFormattedParagraph
               -> (Diagram Pdf R2,Diagram Pdf R2) -- ^ Text and bounding rect of the typeset text
genericPdfText suggested o w h formatted = 
    let diag = mkQD (Prim (PdfTextBox mempty w h formatted))
                    (getEnvelope r)
                    (getTrace r)
                    mempty
                    (Query $ \p -> Any (isInsideEvenOdd p r))
        f v = (moveOriginTo v diag, moveOriginTo v textBounds)
    in
    case o of 
            LeftSide -> f east 
                  where 
                    east = p2 (0,-hlinewrap / 2.0)
            RightSide -> f west 
                  where 
                    west = p2 (wlinewrap,-hlinewrap / 2.0)
            Center -> f theCenter
                  where 
                    theCenter = p2 (wlinewrap / 2.0,-hlinewrap / 2.0)
            TopSide -> f topSide 
                  where 
                    topSide = p2 (wlinewrap / 2.0,0)
            BottomSide -> f bottomSide 
                  where 
                    bottomSide = p2 (wlinewrap / 2.0,-hlinewrap)
            TopLeftCorner -> f topLeft 
                  where 
                    topLeft = p2 (0,0)
            BottomLeftCorner -> f bottomLeft
                  where 
                    bottomLeft = p2 (0,-hlinewrap )
            TopRightCorner -> f topRight
                  where 
                    topRight = p2 (wlinewrap,0)
            BottomRightCorner -> f bottomRight
                  where 
                    bottomRight = p2 (wlinewrap,-hlinewrap )
        
  where wlinewrap :: Double 
        hlinewrap :: Double
        Rectangle (xa :+ ya) (xb :+ yb) | suggested = matchingContainerSize w h formatted
                                        | otherwise = Rectangle (0 :+ 0) (w :+ h)
        wlinewrap = xb - xa
        hlinewrap = yb - ya
   
        r :: Path R2
        r = rect wlinewrap hlinewrap # moveOriginTo (p2 (-wlinewrap / 2.0,hlinewrap / 2.0))
        textBounds :: Diagram Pdf R2
        textBounds = Sh.rect wlinewrap hlinewrap # moveOriginTo (p2 (-wlinewrap / 2.0,hlinewrap / 2.0))

-- | Typeset a text with a given style in a suggested box.
-- The function is returning a diagram for the text
-- and a diagram for the bounding box which may be smaller
-- than the suggested size : smaller width when the algorithm
-- has done some line justification. 
-- The text may also be bigger than the suggested width in case
-- of overflow (similar to the way TeX is doing thing. There are
-- settings in HPDF to control the elegance of the line cuts but
-- those settings are not accessible from this simple API).
-- The text will not be longer than the suggested height. In that
-- case the additional text is not displayed except perhaps partially the last
-- line since no clipping is done.
pdfLabelWithSuggestedSize :: (Renderable PdfTextBox Pdf,Renderable (Path R2) Pdf) 
                          => LabelStyle -- ^ Style of the label
                          -> String -- ^ String to display with this style
                          -> Double -- ^ Suggested width
                          -> Double -- ^ Suggested height
                          -> (Diagram Pdf R2,Diagram Pdf R2) -- ^ Text and bounding rect of the typeset text
pdfLabelWithSuggestedSize (LabelStyle fn fs j o fillc) s w h = 
  let pdfColor (r,g,b,_) = P.Rgb r g b
      pdffc = pdfColor . colorToSRGBA . toAlphaColour $ fillc 
  in
  genericPdfText True o w h $ (AFP NormalParagraph (P.Font (PDFFont fn fs) pdffc pdffc) $ do 
    setJustification j
    paragraph $ do
        txt $ s)

-- | Similar to the @pdfLabelWithSuggestedSize@ but supporting the full features of HPDF
pdfTextWithSuggestedSize :: (ParagraphStyle ps s, P.Style s,Renderable PdfTextBox Pdf,Renderable (Path R2) Pdf) 
                         => TextOrigin -- ^ Text origin
                         -> Double -- ^ Suggested width
                         -> Double -- ^ Suggested height
                         -> ps -- ^ Paragraph (vertical) style
                         -> s  -- ^ Horizontal style
                         -> TM ps s () -- ^ Text
                         -> (Diagram Pdf R2,Diagram Pdf R2) -- ^ Text and bounding rect of the typeset text
pdfTextWithSuggestedSize o w h ps hs tm = genericPdfText True o w h (AFP ps hs tm)

-- | Similar to the @pdfLabelWithSuggestedSize@ but here the size is forced and even
-- if the bounding box of the text is smaller it will not be taken into account
-- for the diagram envelope.
pdfLabelWithSize :: (Renderable PdfTextBox Pdf,Renderable (Path R2) Pdf) 
                 => LabelStyle -- ^ Style of the label
                 -> String -- ^ String to display with this style
                 -> Double -- ^ Suggested width
                 -> Double -- ^ Suggested height
                 -> Diagram Pdf R2 -- ^ Text
pdfLabelWithSize (LabelStyle fn fs j o fillc) s w h = 
  let pdfColor (r,g,b,_) = P.Rgb r g b
      pdffc = pdfColor . colorToSRGBA . toAlphaColour $ fillc 
  in
  fst $ genericPdfText False o w h $ (AFP NormalParagraph (P.Font (PDFFont fn fs) pdffc pdffc) $ do 
    setJustification j
    paragraph $ do
        txt $ s)

-- | Similar to @pdfTextWithSuggestedSize@ but the size if forced and not just suggested
pdfTextWithSize :: (ParagraphStyle ps s, P.Style s,Renderable PdfTextBox Pdf,Renderable (Path R2) Pdf) 
                => TextOrigin -- ^ Text origin
                -> Double -- ^ Suggested width
                -> Double -- ^ Suggested height
                -> ps -- ^ Paragraph (vertical) style
                -> s  -- ^ Horizontal style
                -> TM ps s () -- ^ Text
                -> Diagram Pdf R2 -- ^ Text
pdfTextWithSize o w h ps hs tm = fst $ genericPdfText False o w h (AFP ps hs tm)

instance Renderable PdfImage Pdf where
  render _ (PdfImage t ref) = D $ do
    withContext $ do
       pdfTransf t
       drawM . drawXObject $ ref 

-- | Create an image diagram
pdfImage :: (Monad m, PDFGlobals m)
         => PDFReference PDFJpeg -- ^ Reference to the Jpeg image in the PDF resources
         -> m (Diagram Pdf R2)
pdfImage ref = do 
    (w,h) <- P.bounds ref
    let r :: Path R2
        r = rect w h # moveOriginTo (p2 (-w/2, h/2.0))
        diag = mkQD (Prim (PdfImage mempty ref))
                    (getEnvelope r)
                    (getTrace r)
                    mempty
                    (Query $ \p -> Any (isInsideEvenOdd p r))
    return (diag # moveOriginTo (p2 (w/2.0,h/2.0)))

instance Renderable PdfURL Pdf where
  render _ (PdfURL t url w h) = D $ do
    withContext $ do
       pdfTransf t
       drawM $ do 
        newAnnotation (URLLink (toPDFString "diagrams link") [0,0,w,h] url True)

-- | Create an URL diagram
pdfURL :: String -- ^ URL (in a next version it should be URI)
       -> Double -- ^ Width of active area
       -> Double -- ^ Height of active area
       -> Diagram Pdf R2 
pdfURL url w h = 
  let r = rect w h # moveOriginTo (p2 (-w/2, h/2.0))
      diag = mkQD (Prim (PdfURL mempty url w h))
                    (getEnvelope r)
                    (getTrace r)
                    mempty
                    (Query $ \p -> Any (isInsideEvenOdd p r))
  in 
  diag # moveOriginTo (p2 (w/2.0,h/2.0))


-- To avoid an Orphan instance warning for Transformable PDFShading
newtype TransSh =TransSh {unTrans :: PDFShading}

type instance V TransSh = R2

instance Transformable TransSh where 
    transform t (TransSh (AxialShading xa ya xb yb ca cb)) = TransSh $ AxialShading xa' ya' xb' yb' ca cb 
      where 
        (xa',ya') = unp2 . transform t $ p2 (xa,ya)
        (xb',yb') = unp2 . transform t $ p2 (xb,yb)
    transform t (TransSh (RadialShading xa ya ra xb yb rb ca cb)) = TransSh $ RadialShading xa' ya' ra xb' yb' rb ca cb 
      where 
        (xa',ya') = unp2 . transform t $ p2 (xa,ya)
        (xb',yb') = unp2 . transform t $ p2 (xb,yb)
       
{-

Paragraph shapes 

-}

{-    
data EnvelopedPara v = EnvelopedPara (Envelope R2) v

instance ComparableStyle v => ComparableStyle (EnvelopedPara v) where 
  isSameStyleAs (EnvelopedPara _ va) (EnvelopedPara _ vb) = isSameStyleAs va vb 

instance (ComparableStyle v,P.Style s, ParagraphStyle v s) => ParagraphStyle (EnvelopedPara v) s where 
   linePosition (EnvelopedPara _ v) = P.linePosition v
   lineWidth (EnvelopedPara _ v) = P.lineWidth v
   interline (EnvelopedPara _ v) = P.interline v
   paragraphChange (EnvelopedPara a v) i l = let (np,r) = P.paragraphChange v i l in (EnvelopedPara a np,r)
   paragraphStyle (EnvelopedPara _ v) = P.paragraphStyle v

-}

{-# LANGUAGE DeriveDataTypeable, NoMonomorphismRestriction #-}
-----------------------------------------------------------------------------
-- |
-- Module      :  Diagrams.Backend.Pdf.CmdLine
-- Copyright   :  (c) 2013 alpheccar.org (see LICENSE)
-- License     :  BSD-style (see LICENSE)
--
-- Convenient creation of command-line-driven executables for
-- rendering diagrams using the Pdf backend.
--
-- * 'defaultMain' creates an executable which can render a single
--   diagram at various options.
--
-----------------------------------------------------------------------------

module Diagrams.Backend.Pdf.CmdLine
       ( defaultMain
       , multipleMain
       , Pdf
       ) where

import Diagrams.Prelude hiding (width, height, interval)
import Diagrams.Backend.Pdf

import System.Console.CmdArgs.Implicit hiding (args)

import Prelude

import Data.List.Split

import System.Environment  (getProgName)
import qualified Graphics.PDF as P

data DiagramOpts = DiagramOpts
                   { width     :: Maybe Int
                   , height    :: Maybe Int
                   , output    :: FilePath
                   , compressed       :: Maybe Bool
                   , author :: Maybe String
                   }
  deriving (Show, Data, Typeable)

diagramOpts :: String -> DiagramOpts
diagramOpts prog = DiagramOpts
  { width =  def
             &= typ "INT"
             &= help "Desired width of the output image (default 400)"

  , height = def
             &= typ "INT"
             &= help "Desired height of the output image (default 400)"

  , output = def
           &= typFile
           &= help "Output file"

  , compressed = def
              &= typ "BOOL"
              &= help "Compressed PDF file"
  , author = def 
              &= typ "STRING"
              &= help "Author of the document"
  }
  &= summary "Command-line diagram generation."
  &= program prog

-- | This is the simplest way to render diagrams, and is intended to
--   be used like so:
--
-- > ... other definitions ...
-- > myDiagram = ...
-- >
-- > main = defaultMain myDiagram
--
--   Compiling a source file like the above example will result in an
--   executable which takes command-line options for setting the size,
--   output file, and so on, and renders @myDiagram@ with the
--   specified options.
--
--   Pass @--help@ to the generated executable to see all available
--   options.  Currently it looks something like
--
-- @
-- Command-line diagram generation.
--
-- Foo [OPTIONS]
-- 
-- Common flags:
--   -w --width=INT    Desired width of the output image (default 400)
--   -h --height=INT   Desired height of the output image (default 400)
--   -o --output=FILE  Output file
--   -c --compressed   Compressed PDF file
--   -? --help         Display help message
--   -V --version      Print version information
-- @
--
--   For example, a couple common scenarios include
--
-- @
-- $ ghc --make MyDiagram
--
--   # output image.eps with a width of 400pt (and auto-determined height)
-- $ ./MyDiagram --compressed -o image.pdf -w 400
-- @

defaultMain :: Diagram Pdf R2 -> IO ()
defaultMain d = do
  prog <- getProgName
  opts <- cmdArgs (diagramOpts prog)
  let sizeSpec = case (width opts, height opts) of
                            (Nothing, Nothing) -> Absolute
                            (Just wi, Nothing)  -> Width (fromIntegral wi)
                            (Nothing, Just he)  -> Height (fromIntegral he)
                            (Just wi, Just he)   -> Dims (fromIntegral wi)
                                                       (fromIntegral he)
      (w,h) = sizeFromSpec sizeSpec
      theAuthor = maybe "diagrams-pdf" id (author opts)
      compression = maybe False id (compressed opts)
      docRect = P.PDFRect 0 0 (floor w) (floor h)
      pdfOpts = PdfOptions sizeSpec
  ifCanRender opts $ do 
    P.runPdf (output opts) 
           (P.standardDocInfo { P.author=P.toPDFString theAuthor, P.compressed = compression}) docRect $ do
              page1 <- P.addPage Nothing
              P.drawWithPage page1 $ renderDia Pdf pdfOpts d

-- | Generate a multipage PDF document from several diagrams.
-- Each diagram is scaled to the page size
multipleMain :: [Diagram Pdf R2] -> IO ()
multipleMain d = do
  prog <- getProgName
  opts <- cmdArgs (diagramOpts prog)
  let sizeSpec = case (width opts, height opts) of
                            (Nothing, Nothing) -> Absolute
                            (Just wi, Nothing)  -> Width (fromIntegral wi)
                            (Nothing, Just he)  -> Height (fromIntegral he)
                            (Just wi, Just he)   -> Dims (fromIntegral wi)
                                                       (fromIntegral he)
      (w,h) = sizeFromSpec sizeSpec
      theAuthor = maybe "diagrams-pdf" id (author opts)
      compression = maybe False id (compressed opts)
      docRect = P.PDFRect 0 0 (floor w) (floor h)
      pdfOpts = PdfOptions sizeSpec
      createPage aDiag = do 
        page1 <- P.addPage Nothing
        P.drawWithPage page1 $ renderDia Pdf pdfOpts aDiag 
  ifCanRender opts $ do 
    P.runPdf (output opts) 
           (P.standardDocInfo { P.author=P.toPDFString theAuthor, P.compressed = compression}) docRect $ do
              mapM_ createPage d
              


ifCanRender :: DiagramOpts -> IO () -> IO ()
ifCanRender opts action =
  case splitOn "." (output opts) of
    [""] -> putStrLn "No output file given."
    ps |  last ps `elem` ["pdf"] -> action
       | otherwise -> putStrLn $ "Unknown file type: " ++ last ps

{-# LANGUAGE TypeFamilies              #-}
{-# LANGUAGE MultiParamTypeClasses     #-}
{-# LANGUAGE FlexibleContexts          #-}
{-# LANGUAGE DeriveDataTypeable        #-}
{-# LANGUAGE GeneralizedNewtypeDeriving #-}
{-# LANGUAGE ExistentialQuantification #-}
module Diagrams.Backend.Pdf.Specific(
      LabelStyle(..)
    , PdfTextBox(..)
    , drawStringLabel
    , getTextBoundingBox
    , TextOrigin(..)
    , LabelSize
    , PdfImage(..)
    , PdfURL(..)
    , PdfShadingData(..)
    , getShadingData
    , pdfAxialShading
    , pdfRadialShading
    , AnyFormattedParagraph(..)
    , CanBeFormatted(..)
    ) where 

import Graphics.PDF hiding(translate)
import qualified Graphics.PDF as P
import Diagrams.Prelude
import Data.Typeable

--import qualified Debug.Trace as T 

--debug a = T.trace (show a) a

data TextOrigin = Center 
                | LeftSide 
                | RightSide 
                | TopSide 
                | BottomSide 
                | TopLeftCorner 
                | TopRightCorner 
                | BottomLeftCorner 
                | BottomRightCorner 
                deriving(Eq)

type LabelSize = Int

-- | Style for a label.
-- It is not considered as an attribute but as a different primitive
-- because internaly it is a complex text which can support several styles 
-- in the same paragraph. 
-- Label is just a convenience wrapper when the full features are not needed
data LabelStyle = LabelStyle FontName LabelSize Justification TextOrigin (Colour Double)

data PdfTextBox = PdfTextBox { _transform :: T2 
                             , _suggestedWidth :: Double 
                             , _suggestedHeight :: Double 
                             , _paragraph :: AnyFormattedParagraph
                             }

type instance V PdfTextBox = R2

instance Transformable PdfTextBox where
  transform t (PdfTextBox tt sw sh para) = PdfTextBox (t <> tt) sw sh para

instance IsPrim PdfTextBox

instance HasOrigin PdfTextBox where
  moveOriginTo p = translate (origin .-. p)

instance Renderable PdfTextBox NullBackend where
  render _ _ = mempty

data PdfImage = PdfImage T2 (PDFReference PDFJpeg)

type instance V PdfImage = R2

instance Transformable PdfImage where
  transform t (PdfImage tt ref) = PdfImage (t <> tt) ref

instance IsPrim PdfImage

instance HasOrigin PdfImage where
  moveOriginTo p = translate (origin .-. p)

instance Renderable PdfImage NullBackend where
  render _ _ = mempty

data PdfURL = PdfURL T2 String Double Double 

type instance V PdfURL = R2

instance Transformable PdfURL where
  transform t (PdfURL tt s w h) = PdfURL (t <> tt) s w h

instance IsPrim PdfURL

instance HasOrigin PdfURL where
  moveOriginTo p = translate (origin .-. p)

instance Renderable PdfURL NullBackend where
  render _ _ = mempty

drawStringLabel :: PDFFloat 
                -> PDFFloat 
                -> AnyFormattedParagraph
                -> Draw () 
drawStringLabel w h para = typesetText w h para

data AnyFormattedParagraph = forall s ps. (ParagraphStyle ps s, P.Style s) => AFP ps s (TM ps s ())

class CanBeFormatted m where 
  putIntoContainer :: Double -> Double -> m -> Draw ()
  matchingContainerSize :: Double -> Double -> m -> Rectangle

instance CanBeFormatted AnyFormattedParagraph where 
  putIntoContainer w h (AFP ps p t) = 
    let b = getBoxes ps p t
        sh = styleHeight p
        c = mkContainer 0 0 w h sh
        (d,_,_) = fillContainer (defaultVerState ps) c b
    in 
    d
  matchingContainerSize w h (AFP ps p t) = 
    let b = getBoxes ps p t
        sh = styleHeight p
        c = mkContainer 0 0 w h sh
        (_,c',_) = fillContainer (defaultVerState ps) c b
    in 
    containerContentRectangle  c'

typesetText :: PDFFloat -- ^ width limit
            -> PDFFloat -- ^ height limit
            -> AnyFormattedParagraph
            -> P.Draw ()
typesetText w h para = putIntoContainer w h para

getTextBoundingBox :: PDFFloat -- ^ width limit
                   -> PDFFloat -- ^ height limit
                   -> AnyFormattedParagraph
                   -> Rectangle
getTextBoundingBox w h para  = matchingContainerSize w h para

data PdfShadingData = PdfAxialShadingData P2 P2 (Colour Double) (Colour Double) 
                    | PdfRadialShadingData P2 Double P2 Double (Colour Double) (Colour Double) deriving (Show,Typeable)
newtype PdfShading = PdfShading (Last PdfShadingData) deriving (Typeable, Semigroup)

instance AttributeClass PdfShading

getShadingData :: PdfShading -> PdfShadingData
getShadingData (PdfShading (Last c)) = c

addshading :: (HasStyle a) => PdfShadingData -> a -> a
addshading s = applyAttr (PdfShading . Last $ s)


-- | Define Axial shading for a diagram
pdfAxialShading :: HasStyle a => P2 -> P2 -> Colour Double -> Colour Double -> a -> a
pdfAxialShading pa pb ca cb = addshading (PdfAxialShadingData pa pb ca cb)

-- | Define Radial shading for a diagram
pdfRadialShading :: HasStyle a 
                 => P2 -- ^ Center of inner circle
                 -> Double -- ^ Radius of inner circle
                 -> P2 -- ^ Center of outer circle
                 -> Double -- ^ Radius of outer circle
                 -> Colour Double -- ^ Inner colour
                 -> Colour Double -- ^ Outer colour
                 -> a -> a
pdfRadialShading pa ra pb rb ca cb = addshading (PdfRadialShadingData pa ra pb rb ca cb)

all:
	ghc -o mytest --make test.hs -package-db ../dist/package.conf.inplace 

clean:
	rm -f mytest 
	rm -f *.o 
	rm -f *.hi
	rm -f *.pdf

run:
	./mytest 

{-# LANGUAGE NoMonomorphismRestriction #-}
{-# LANGUAGE ScopedTypeVariables #-}
{-# LANGUAGE MultiParamTypeClasses #-}
import Diagrams.Prelude
import Diagrams.Backend.Pdf
import Diagrams.Backend.Pdf.CmdLine
import Data.Colour (withOpacity)
import Graphics.PDF hiding(scale,red,green,blue,white,black,text,rotate,rect,stroke)
import qualified Diagrams.Backend.SVG.CmdLine as S
import qualified Diagrams.Example.Logo as L
import           Diagrams.Coordinates ((&))
import qualified Graphics.PDF.Typesetting as T
import qualified Graphics.PDF as P
import System.Random

pageWidth = 600 
pageHeight = 400 
titleH = 50

data MyParaStyles = Normal
                  | Bold
                  | Crazy
                  | SuperCrazy [Int] [PDFFloat]
                  | DebugStyle
                  | RedRectStyle
                  | BlueStyle
                  
instance ComparableStyle MyParaStyles where
  isSameStyleAs Normal Normal = True
  isSameStyleAs Bold Bold = True
  isSameStyleAs Crazy Crazy = True
  isSameStyleAs (SuperCrazy _ _) (SuperCrazy _ _) = True
  isSameStyleAs DebugStyle DebugStyle = True
  isSameStyleAs RedRectStyle RedRectStyle = True
  isSameStyleAs BlueStyle BlueStyle = True
  isSameStyleAs _ _ = False
  
                  
instance T.Style MyParaStyles where
    textStyle Normal = TextStyle (PDFFont Times_Roman 10) P.black P.black FillText 1.0 1.0 1.0 1.0
    textStyle Bold = TextStyle (PDFFont Times_Bold 12) P.black P.black FillText 1.0 1.0 1.0 1.0
    textStyle RedRectStyle = TextStyle (PDFFont Times_Roman 10) P.black P.black FillText 1.0 1.0 1.0 1.0
    textStyle DebugStyle = TextStyle (PDFFont Times_Roman 10) P.black P.black FillText 1.0 1.0 1.0 1.0
    textStyle Crazy = TextStyle (PDFFont Times_Roman 10) P.red P.red FillText 1.0 1.0 1.0 1.0
    textStyle (SuperCrazy _ _) = TextStyle (PDFFont Times_Roman 12) P.black P.black FillText 1.0 2.0 0.5 0.5
    textStyle BlueStyle = TextStyle (PDFFont Times_Roman 10) P.black P.black FillText 1.0 1.0 1.0 1.0
    
    sentenceStyle BlueStyle = Just $ \r d -> do
        P.fillColor $ Rgb 0.6 0.6 1
        P.strokeColor $ Rgb 0.6 0.6 1
        P.fillAndStroke r
        d
        return()
    
    sentenceStyle RedRectStyle = Just $ \r d -> do
        P.strokeColor P.red
        P.stroke r
        d
        return()
    sentenceStyle Crazy = Just $ \r d -> do
       d
       P.strokeColor P.blue
       P.stroke r
    sentenceStyle _ = Nothing
           
    wordStyle DebugStyle = Just $ \r m d ->
      case m of
          DrawWord -> d >> return ()
          DrawGlue -> d >> P.stroke r
    wordStyle Crazy = Just crazyWord
    wordStyle (SuperCrazy l _) = Just ws 
     where
        ws _ DrawGlue _ = return ()
        ws (Rectangle (xa :+ ya) (xb :+ yb)) DrawWord drawWord = do
            let [a,b,c,d,e,f,g,h] :: [PDFFloat] = map (\x -> x / 16.0) . map fromIntegral . take 8 $ l
                --angle = head angl
                p = Polygon [ (xa-a) :+ (ya+b)
                            , (xb+c) :+ (ya+d)
                            , (xb+e) :+ (yb-f)
                            , (xa-g) :+ (yb-h)
                            , (xa-a) :+ (ya+b)
                            ]
            P.strokeColor P.red
            P.stroke p
            P.fillColor $ Rgb 0.8 1.0 0.8
            P.fill p
            withNewContext $ do
              --applyMatrix . rotate . Degree $ angle
              drawWord
            return ()

    wordStyle _ = Nothing
    
    updateStyle (SuperCrazy a b) = SuperCrazy (drop 8 a) (tail b)
    updateStyle a = a
    
    styleHeight r@(SuperCrazy _ _) = (getHeight . textFont . textStyle $ r) + 4.0
    styleHeight r = getHeight . textFont . textStyle $ r
    
    styleDescent r@(SuperCrazy _ _) = (getDescent . textFont . textStyle $ r) + 2
    styleDescent r = getDescent . textFont . textStyle $ r
    
crazyWord :: Rectangle -> StyleFunction -> Draw a -> Draw ()
crazyWord r@(Rectangle (xa :+ ya) (xb :+ yb)) DrawWord d = do
    P.fillColor $ Rgb 0.6 1 0.6 
    P.fill r
    d
    P.strokeColor $ Rgb 0 0 1
    let m = (ya+yb)/2.0
    P.stroke $ P.Line xa m xb m 
crazyWord (Rectangle (xa :+ ya) (xb :+ yb)) DrawGlue _ = do
    P.fillColor $ Rgb 0 0 1
    P.fill (Circle ((xa+xb)/2.0) ((ya+yb)/2.0) ((xb-xa)/2.0))
    
   
    
superCrazy :: MyParaStyles
superCrazy = SuperCrazy (randomRs (0,32) (mkStdGen 0)) (randomRs (-10.0,10.0) (mkStdGen 10000))
    
data MyVertStyles = NormalPara
                  | CirclePara
                  | BluePara !PDFFloat

instance ComparableStyle MyVertStyles where
    isSameStyleAs NormalPara NormalPara = True
    isSameStyleAs CirclePara CirclePara = True
    isSameStyleAs (BluePara _) (BluePara _) = True
    isSameStyleAs _ _ = False


instance ParagraphStyle MyVertStyles MyParaStyles  where
    lineWidth (BluePara a) w nb = (if nb > 3 then w else w-a) - 20.0
    lineWidth CirclePara _ nb = 
           let nbLines = 15.0
               r = nbLines * (getHeight . textFont . textStyle $ Normal)
               pasin x' = if x' >= 1.0 then pi/2 else if x' <= -1.0 then (-pi/2) else asin x'
               angle l = pasin $ (nbLines - (fromIntegral l)  ) / nbLines
           in
           abs(2*r*cos (angle nb))
    lineWidth _ w _ = w
           
    linePosition (BluePara a) _ nb = (if nb > 3 then 0.0 else a) + 10.0
    linePosition a@(CirclePara) w nb = max 0 ((w - P.lineWidth a w nb) / 2.0)
    linePosition _ _ _ = 0.0
    
    interline (BluePara _) = Just $ \r -> do
        P.fillColor $ Rgb 0.6 0.6 1
        P.strokeColor $ Rgb 0.6 0.6 1
        P.fillAndStroke r
    interline _ = Nothing
        
    paragraphChange (BluePara _) _ (AChar st c _:l) = 
        let f = PDFFont Helvetica_Bold 45
            w' = charWidth f c 
            charRect = Rectangle (0 :+ (- getDescent f)) (w' :+ (getHeight f - getDescent f))
            c' = mkLetter (0,0,0) Nothing . mkDrawBox $ do
                withNewContext $ do
                    applyMatrix $ P.translate ((-w') :+ (getDescent f - getHeight f + styleHeight st - styleDescent st))
                    P.fillColor $ Rgb 0.6 0.6 1
                    P.strokeColor $ Rgb 0.6 0.6 1
                    P.fillAndStroke $ charRect
                    P.fillColor P.black
                    drawText $ do
                        renderMode AddToClip
                        textStart 0 0
                        setFont f
                        displayText (toPDFString [c])
                    paintWithShading (AxialShading 0 (- getDescent f) w' (getHeight f - getDescent f) (Rgb 1 0 0) (Rgb 0 0 1)) (addShape charRect)
        in
        (BluePara w', c':l)
    
    paragraphChange s _ l = (s,l)
    
    paragraphStyle (BluePara _) = Just $ \(Rectangle (xa :+ ya) (xb :+ yb)) b -> do
        let f = Rectangle ((xa-3) :+ (ya-3)) ((xb+3) :+ (yb+3))
        P.fillColor $ Rgb 0.6 0.6 1
        P.fill f
        b
        P.strokeColor P.red
        P.stroke f
        return ()
    paragraphStyle _ = Nothing
 
standardStyleTest :: TM MyVertStyles MyParaStyles ()
standardStyleTest = do
    paragraph $ do
        txt $ "Lorem ipsum dolor sit amet, consectetur adipisicing elit, sed do eiusmod tempor incididunt ut "
        txt $ "labore et dolore magna aliqua. Ut enim ad minim veniam, quis nostrud exercitation ullamco laboris "
        setStyle Crazy
        txt $ "nisi ut aliquip ex ea commodo consequat. Duis aute irure dolor in reprehenderit in voluptate "
        txt $ "velit esse cillum dolore eu fugiat nulla pariatur. Excepteur sint occaecat cupidatat non "
        txt $ "proident, sunt in culpa qui officia deserunt mollit anim id est laborum."
 

complexText = 
  centerXY (fst (pdfTextWithSuggestedSize Center 400 200 NormalPara Normal standardStyleTest) 
    ===
    strutY 20 
    ===
    fst (pdfTextWithSuggestedSize Center 300 200 NormalPara Normal standardStyleTest) # rotate (20 :: Deg)
  )
  <> rect pageWidth (pageHeight - titleH)

t s x j = 
    let (td, rd) = pdfLabelWithSuggestedSize (LabelStyle Times_Roman 12 j x blue) s 50 100 
    in 
    td # showOrigin # lw 0.03  <> rd

tfs s x j = 
    let td = pdfLabelWithSize (LabelStyle Times_Roman 12 j x blue) s 50 50 
    in 
    td # showOrigin # lw 0.03 

testpdfsuggestedtextsize = centerXY $ (centerXY squareText) <> rect pageWidth (pageHeight - titleH)
 where 
  squareText = (t "Top Left" TopLeftCorner LeftJustification ||| t "Top" TopSide Centered ||| t "Top Right" TopRightCorner RightJustification)
               ===
               (t "Left" LeftSide LeftJustification ||| t "Center" Center Centered ||| t "Right" RightSide RightJustification)
               ===
               (t "Bottom Left" BottomLeftCorner LeftJustification ||| t "Bottom" BottomSide Centered ||| t "Bottom Right" BottomRightCorner RightJustification)
  
testpdftextsize = centerXY $ (centerXY squareText) <> rect pageWidth (pageHeight - titleH)
 where 
  squareText = (tfs "Top Left" TopLeftCorner LeftJustification ||| tfs "Top" TopSide Centered ||| tfs "Top Right" TopRightCorner RightJustification)
               ===
               (tfs "Left" LeftSide LeftJustification ||| tfs "Center" Center Centered ||| tfs "Right" RightSide RightJustification)
               ===
               (tfs "Bottom Left" BottomLeftCorner LeftJustification ||| tfs "Bottom" BottomSide Centered ||| tfs "Bottom Right" BottomRightCorner RightJustification)
  

 
testShading = 
   let loopyStar = fc red
                 . mconcat . map (cubicSpline True)
                 . pathVertices
                 . star (StarSkip 3)
                 $ regPoly 7 1
       f z d = 
        let s = 20
        in
        loopyStar # scale s # fillRule z # pdfAxialShading (p2 (-1,-1)) (p2 (1,1)) red green # rotate (d :: Deg)
   in   centerXY (hcat (map (f Winding) [0,20,40,60,80,100,120,140,160]))
        === centerXY (hcat (map (f EvenOdd) [0,20,40,60,80,100,120,140,160]))
        === square 40 # pdfRadialShading (p2 (0,0)) 5 (p2 (0,0)) 40 blue red

testImage img = 
    let url = "http://www.alpheccar.org" 
    in 
       circle 100 
    <> mconcat (map (\r -> img # scale 0.5 # translateX 100 # rotate r) ([0,20..360] :: [Deg]))
    <> pdfURL url 100 20 
    <> fst (pdfLabelWithSuggestedSize (LabelStyle Times_Roman 12 Centered Center blue) url 150 40)

mkSection s sect = 
    let (d,_) = pdfLabelWithSuggestedSize (LabelStyle Times_Roman 36 Centered Center blue) s pageWidth pageHeight 
    in do
      page1 <- addPage Nothing
      drawWithPage page1 $ do
        renderDia Pdf (PdfOptions (Dims pageWidth pageHeight)) d
      sect

page s d = header s === content d 
 where 
  header s = 
    let (d,_) = pdfLabelWithSuggestedSize (LabelStyle Times_Roman 24 Centered Center black) s pageWidth titleH 
    in 
    d <> (rect pageWidth titleH # pdfAxialShading (p2 (-pageWidth/2,-titleH/2)) (p2 (pageWidth/2,titleH/2)) blue white)
  clipRect :: Path R2
  clipRect =  rect pageWidth (pageHeight - titleH) 
  clipRectDiag :: Diagram Pdf R2 
  clipRectDiag = stroke clipRect
  content d = (clipRectDiag # lw 0 <> (centerXY d)) # clipBy clipRect # withEnvelope clipRectDiag
  

mkPage :: String -> Diagram Pdf R2 -> PDF ()
mkPage s d = do 
  page1 <- addPage Nothing
  drawWithPage page1 $ do
        renderDia Pdf (PdfOptions (Dims pageWidth pageHeight)) $ page s d

testShadeFroz = w
  where 
    w :: Diagram Pdf R2 
    w = rect pageWidth (pageHeight - titleH) <> centerXY (
       square 100 # pdfAxialShading (p2 (-50,-50)) (p2 (50,50)) blue white # lw 1.0
       ===
       square 50 # freeze # rotate (45 :: Deg) # scale 2 # pdfAxialShading (p2 (-40,-40)) (p2 (40,40)) blue yellow # lw 1.0
    
        )

testOpacity = do 
  page1 <- addPage Nothing
  drawWithPage page1 $ do
    renderDia Pdf (PdfOptions (Dims pageWidth pageHeight)) $ w
  where 
      w :: Diagram Pdf R2 
      w = 
        let colors  = map (blue `withOpacity`) [0.1, 0.2 .. 1.0]
            colorss  = map (black `withOpacity`) [0.1, 0.2 .. 1.0]
        in  hcat' with { catMethod = Distrib, sep = 1 }
                  (zipWith3 (\cs cl c -> fcA cs $ c) colors colorss (repeat (circle 1)))
  
main = do
  Right jpgf <- readJpegFile "logo.jpg" 
  let theDocRect = PDFRect 0 0 pageWidth pageHeight
  runPdf "demo.pdf" (standardDocInfo { author=toPDFString "alpheccar", compressed = False}) theDocRect $ do
      mkSection "HPDF Specific Primitives" $ do
           jpg <- createPDFJpeg jpgf
           image <- pdfImage jpg
           mkPage "Test JPEG and URL" (testImage image)
           mkPage "Test Shading" testShading
           mkPage "Test Suggested Text Container Size" testpdfsuggestedtextsize
           mkPage "Test Forced Text Container Size" testpdftextsize
           mkPage "Text Complex Text" complexText
           mkPage "Test shading with freeze" testShadeFroz
--






